<?xml version="1.0" encoding="UTF-8"?>

<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
	
	<xsl:include href="../../admin.xsl" />
	<xsl:include href="../../common/form.xsl" />
	<xsl:include href="../../cp/path.xsl" />
	
	<!-- Запускаем рендер страницы -->
	<xsl:template match="/data">
		<html>
		
		<head>
			<title>
				<xsl:value-of select="$pageTitle" />
			</title>
			<xsl:call-template name="meta-head" />
			<link rel="stylesheet" href="{$baseURI}/css/admin/forms.css" type="text/css" media="all" />
		</head>
		<body>
			<xsl:call-template name="admin-header" />
			
			<div id="blocks">
				<div id="blocks_inner" />
			</div>
			
			<div id="content">
				<div id="content_inner">
					<xsl:apply-templates select="response" />
				</div>
			</div>
			
			<xsl:call-template name="footer" />
		</body>
		</html>
	</xsl:template>
	
	<xsl:template match="response">
		<div id="path">
			<xsl:apply-templates select="document('xmodule://cp/getPath/0;system;rss')/data/response" mode="cp-path">
				<xsl:with-param name="activeLink">1</xsl:with-param>
			</xsl:apply-templates>
		</div>
		
		<h1>
			<xsl:if test="$method = 'add_rss'">
				<xsl:text>Добавление типа данных</xsl:text>
			</xsl:if>
			<xsl:if test="$method = 'edit_rss'">
				<xsl:text>Редактирование информации о типе данных</xsl:text>
			</xsl:if>
		</h1>
		
		<form action="{$baseURI}/admin/system/add_rss/save" method="post">
			<xsl:if test="$method = 'edit_rss'">
				<xsl:attribute name="action">
					<xsl:value-of select="concat($baseURI, '/admin/system/edit_rss/', $formPrefix, '/save')" />
				</xsl:attribute>
			</xsl:if>
			
			<xsl:apply-templates select="form/field">
				<xsl:with-param name="typeId" select="form/field[@name='type_id']/value" />
			</xsl:apply-templates>
			
			<p>
				<input type="submit" value="Добавить">
					<xsl:if test="$method = 'edit_rss'">
						<xsl:attribute name="action">Сохранить</xsl:attribute>
					</xsl:if>
				</input>
			</p>
		</form>
		
		<script type="text/javascript">
			function reloadFields (typeId) {
				$.ajax ({
					url: baseURI + '/admin/system/ajax_get_type_fields/' + typeId,
					dataType: 'xml',
					type: 'get',
					error: function() { alert('Произошла ошибка!') },
					success: function(xml) {
						$('#field_title_field option[value!=0], #field_description_field option[value!=0], #field_owner_name_field option[value!=0], #field_pubdate_field option[value!=0], #field_tags_field option[value!=0], #field_image_field option[value!=0]').remove();
						
						var fields = $(xml).find('field').each (function() {
							$("<option></option>")
								.attr('value', $(this).find('id').text())
								.text($(this).find('title').text())
								.appendTo('#field_title_field, #field_description_field, #field_owner_name_field, #field_pubdate_field, #field_tags_field, #field_image_field');
						});
						
						$('#field_title_field, #field_description_field, #field_owner_name_field, #field_pubdate_field, #field_tags_field, #field_image_field').attr('disabled',false);
					}
				});
			}
		</script>
	</xsl:template>
	
	<xsl:template match="field[@name='type_id']">
		<xsl:param name="typeId" />
		<xsl:param name="value" select="value" />
		
		<div class="form_field">
			<div class="form_field_label">
				<label for="field_{@id}">
					<xsl:value-of select="title" />
					<xsl:text>:</xsl:text>
				</label>
			</div>
			<div class="form_field_input">
				<select id="field_{@id}" class="form_input_select" name="data[{$formPrefix}][{@name}]" onchange="reloadFields(this.value)">
					<option value="">-- выберите</option>
					
					<xsl:for-each select="document('xmodule://system/getTypes')/data/response/object[@type='explay_type']">
						<option value="{@id}">
							<xsl:if test="$value = @id">
								<xsl:attribute name="selected">selected</xsl:attribute>
							</xsl:if>
							
							<xsl:value-of select="property[@name='title']/value" />
						</option>
					</xsl:for-each>				
				</select>
				<xsl:if test="error != ''">
					<br /><span class="form_field_error"><xsl:value-of select="error" /></span>
				</xsl:if>
				<xsl:if test="help != ''">
					<br /><span class="form_field_help"><xsl:value-of select="help" /></span>
				</xsl:if>
			</div>
		</div>
	</xsl:template>
	
	<xsl:template match="field[@name='title_field' or @name='description_field' or @name='owner_name_field' or @name='pubdate_field' or @name='tags_field' or @name='image_field']">
		<xsl:param name="typeId" />
		<xsl:param name="value" select="value" />
		
		<div class="form_field">
			<div class="form_field_label">
				<label for="field_{@id}">
					<xsl:value-of select="title" />
					<xsl:text>:</xsl:text>
				</label>
			</div>
			<div class="form_field_input">
				<select id="field_{@name}" class="form_input_select" name="data[{$formPrefix}][{@name}]">
					<xsl:if test="$typeId = ''">
						<xsl:attribute name="disabled">disabled</xsl:attribute>
					</xsl:if>
					
					<option value="0">-- выберите</option>
					
					<xsl:if test="$typeId != ''">
						<xsl:for-each select="document(concat('xmodule://system/getTypeFields/', $typeId))/data/response/object[@type='explay_field']">
							<option value="{@id}">
								<xsl:if test="$value = @id">
									<xsl:attribute name="selected">selected</xsl:attribute>
								</xsl:if>
								
								<xsl:value-of select="property[@name='title']/value" />
							</option>
						</xsl:for-each>
					</xsl:if>
				</select>
				<xsl:if test="error != ''">
					<br /><span class="form_field_error"><xsl:value-of select="error" /></span>
				</xsl:if>
				<xsl:if test="help != ''">
					<br /><span class="form_field_help"><xsl:value-of select="help" /></span>
				</xsl:if>
			</div>
		</div>
	</xsl:template>

	
</xsl:stylesheet>